Module 2.  Advanced Data manipulation
(i) Aggregating  (collapsing)  the data

To aggregate data the break variable(s)  should be chosen as well as names of new variables to  be aggregated and the aggregate functions (such as count, sum, mean, standard deviation etc.) for each one.

For example, using MEMBERS.sav file let’s create the new variable hsize – the number of all persons for each household.

We need to open the file MEMBERS.sav:  

GET  FILE=‘D:\_WORK\Serbia\MEMBERS.sav'.







Choose from the menu:

Data


Aggregate

Select hhid in Break Variable(s) box. To compute the number of all rows for each  household  select the  Save number of cases in break group as variable  option. We can rename the new variable name N_BREAK  to  hsize. 

Press Paste to copy the following command in the syntax window:

AGGREGATE

  /OUTFILE=‘D:\_WORK\Serbia\AGGRtemp.sav'

  /BREAK=hhid
  /hsize=N.

As a result the new SPSS data file by default named AGGRtemp.sav (could be renamed) with 2 variables ( hhid and  aggregated hsize) will be created. 

Other aggregate variables could be created using other aggregate functions, such as mean, min, max, sum, standard deviation, etc.  (from Function box). 

(ii)  Merging files containing different variables

Often we need to combine the different datasets with different variables and the same or different  number of observations. Before we can combine two files containing different information, we need to sort the observations by the same variable in both datasets. Usually the same variable (key variable) shared by the datasets represent an identifier of the observations. 
In our data, for example, files HH.sav and AGGRtemp.sav have the variable hhid (Code of the household) as a key. These files have the same number of observations.

Before merging these two files, we would have to sort the observations in both of them by hhid.

First, open the data file HH.sav, sort it by hhid and save sorted file with the same name:

GET  FILE=‘D:\_WORK\Serbia\HH.sav'.






SORT CASES BY   hhid .










SAVE OUTFILE=‘D:\_WORK\Serbia\HH.sav'

   /COMPRESSED.


Now we have to perform the same procedure on AGGRtemp.sav.





GET  FILE=‘D:\_WORK\Serbia\AGGRtemp.sav'.






SORT CASES BY   hhid .










SAVE OUTFILE=‘D:\_WORK\Serbia\AGGRtemp.sav'

   /COMPRESSED.


After we sorted and saved both files by the same variable, we can merge them. We need to open one of the files we want to merge, for instance HH.sav:  

GET  FILE=‘D:\_WORK\Serbia\HH.sav'.








Second, to add the variables from the external file AGGRtemp.sav to the working file HH.sav, choose from the menu:

Data


Merge Files




Add variables
Enter the name of the file AGGRtemp.sav that we want to merge with HH.sav. To identify and correctly match observations from the two files according to the family code variable, select Match cases on key variables in sorted files and  hhid in the Key variables box. 

We can also select the Indicate case source as variable option. This creates an indicator variable that indicates the source data file for each case in the merged file: 

· For observations from the working file HH.sav that are missing in the external file AGGRtemp.sav, the value of this variable is 0;

· For observations from the external data file that are missing from the working file and for observations present in both files, the value of this variable is 1. 
The default name of this variable is source01.
Press Paste to copy the following command in the syntax window:

MATCH FILES 

 /FILE=*

 /FILE='D:\_WORK\Serbia\AGGRtemp.sav'

 /IN=source01

 /BY hhid .

EXECUTE.

This command tells SPSS to use the working file HH.sav and to add all the variables from the external file AGGRtemp.sav. 

To see the missing in the external or working file use the command:

FREQ source01.
NOTE: the MATCH FILES command allows to merge files with the same number of observations. If we need to merge files with different number of observations we need to use sub-command TABLE.
For example, if we want to merge a file with household data with a file with individual data using the household code variable as a key (hhid), we have to specify which file provides a “table lookup” (or “keyed table”) for the new merged file. 

We need to open MEMBERS.sav as working file. To add the variables from the external file HH.sav, choose from the menu:  

Data


Merge Files




Add variables
and enter the name of the file HH.sav. 
Select Match cases on key variables in sorted files  and choose External file is keyed table. Select  hhid in the Key variables box. 

In this case the merge operation will be as follows:

 MATCH FILES 
 /FILE=*

 /TABLE=‘D:\_WORK\Serbia\HH.sav'

 /BY hhid.

EXECUTE.
This command tells SPSS to use the working file MEMBERS.sav and to add all the variables from the external file HH.sav. The number of  observations in the working and new files is the same.
Follow-up practice (a)
1. Based on the file MEMBERS.sav  create a new file with some characteristics of household head (for head of household relation=1) and produce a frequency distribution tables on sex, age (age grouped), marital status, education level, activity status and occupation of household heads. (Do weight the data.)

2. Create a new file with some aggregated variables:  the number of all persons in each household,  the count of men and women in the household, number of children under 5 years and under 15 years in the household and  produce means of their.  

3. Compute the number of employed and number of unemployed in the households and compare average number of employed and unemployed among households in urban and rural areas (type), in different regions (region), among households male and female headed.
4. Compare average number of children among households in urban and rural areas, in different regions, among households male and female headed.

5. Compare the average household size and  average age of head among households male and female headed.
6. Calculate average monthly per capita income and report the mean and median per capita income for each educational category of individual.  
7. Report mean and median per capita income for people of households male and female headed.

8. Report mean and median per capita income for people of age groups 0-17, 18-65, 65 and more.

9. Report average age of household head, gender ratio (% of female), % of employed household head, % of higher education household head  - for each Deciles of  per capita income.

Practical application:
(A)   Creating Household composition variables

Based on the file MEMBERS.sav  we need to create the new file HH_composition.sav,  that will include some new variables: household size, number of children under 5 years, number of children under 14 years and number of children under 18 years, number of adult age 14 and more, number of adult  age 18 and more, number of elderly (age 65 and more). 
Also we will create a few variables for household head characteristics: sex, age, education level etc. of household head. 
We need also prepare some equivalence scales to account for household composition. Add another measure of household size, which takes into account the fact that children have smaller needs than adults.  More precisely the OECD and OECD modified scale variables will be created (AEoecd and AEoecdm).

OECD scale assumes that the first adult’s weight is 1, other adults have weight 0.7 and the children  have the weight 0.5.

OECD modified scale assumes that the first adult’s weight is 1, other adults have weight 0.5 and the children have the weight 0.3.

GET FILE  'D:\_WORK\Serbia\MEMBERS.sav' .

if  (age<5)     child5  = 1.

if  (age<14)   child14 = 1.

if  (age<18)   child18 = 1.

if  (age>=14)  adult14 = 1.

if  (age>=18)  adult18 = 1.

if  (age>=65)  elderly = 1.

EXEC.
AGGR OUTFILE  *   /BREAK hhid 

/hhsize  'Household size' = n

/child5  'No. of children under age 5'  = sum (child5)

/child14 'No. of children under age 14' = sum (child14)

/child15 'No. of children under age 18' = sum (child18)

/adult14 'No. of adult age 14 and more' = sum (adult14) 

/adult15 'No. of adult age 18 and more' = sum (adult18) 

/elderly 'No. of elderly (age 65 and more)' = sum (elderly) 

/hsex    'HH head, Sex' = first (sex) 

/hage    'HH head, Age' = first (age)

/heduc   'HH head, Education level' = first (educ).

EXEC.

RECODE adult14 adult15 child5 child14  child15 elderly  (sysmis = 0).

comp AEoecd  = 1+ (adult14-1)*0.7 + 0.5*child14.

comp AEoecdm = 1+ (adult14-1)*0.5 + 0.3*child14.

var lab AEoecd  "OECD adult equivalent size of HH".

var lab AEoecdm "Modified OECD adult equivalent size of HH".

FORMATs hhsize AEoecd AEoecdm (F8.2).

MEANS  AEoecd AEoecdm by hhsize /cells mean.

SAVE OUTFILE  'D:\_WORK\Serbia\HH_composition.sav' . 

(B) Working with  DIARY.sav  file
1. Creating the prices of food products by regions  and urban/rural location.

Based on the file DIARY.sav  we need to create the new file PRICES.sav which is the average and median prices and number of observations of each food item by canton and urban/rural location.

Please note that the structure of  so called “long diaries” such as the file DIARY.sav (where each food item is recorded as a separate case in the dataset in the contrary to so called “wide diaries”, where each food is recorded as a separate variable in the dataset)  is much more convenient for some data manipulations like calculating price.
GET FILE  DIARY.sav .

COMPUTE price = PAID_IN/ QUANTITY.

AGGR OUTFILE  *

/BREAK Region  UR  product produc_a coicop_g

/ avgprice  = mean (price)

/ medprice = median (price)

/ Nobserv  = n.

VAR LAB 

 avgprice 'Average price'

/medprice 'Median price'

/nobserv  'Number of observations'.

SAVE OUTFILE 'PRICES.sav' .

2. Missing quantities and values.

Replace the missing quantities in the DIARY.sav  file by the imputed quantity, which is the amount paid divided by the median price of the product in particular canton and location area (urban/ rural). Replace the missing values by the imputed value, which is the quantity divided by the median price of the product in particular canton and location area.
GET FILE  DIARY.sav .

SORT CASES BY region product .

MATCH FILES

 /FILE *

 /TABLE  PRICES.sav

 /BY region product.

EXECUTE.

if  (  sysmis (quantity)  and  not( sysmis(paid_in) )  ) quantity = paid_in/medprice.

if  (  sysmis (paid_in)  and  not( sysmis(quantity) )  ) paid_in = quantity *medprice.

EXECUTE.

SAVE OUTFILE DIARY_with_prices_temp.sav .

3. Computing the household total monthly food expenditures. 

Based on the file DIARY.sav  we need to create the new file: HH_Total_consumption.sav (household level file),  that along with hhid variable will include one new variable: household total monthly food expenditures.

GET FILE  FOOD_DIARY_with_prices_temp.sav' .

AGGR OUTFILE  *

/BREAK hhid

/totexp = sum(paid_in).

VAR LAB  totexp 'Household total monthly expenditures'.

SAVE OUTFILE  'D:\_WORK\Serbia\EXPtemp.sav' .

Follow-up practice (b)

1. Create population deciles based on per capita and per adult equivalent consumption.
Cross-tabulate the population deciles of per capita consumption by the population deciles of per adult equivalent consumption. Comment the table.
2. Create household deciles based on per capita consumption.

Cross-tabulate the household deciles of per capita consumption by the population deciles of per capita consumption. Comment the table.
3. Compute Average per capita consumption and Average per adult equivalent consumption:


 

- by cantons/districts;


- by gender of the head of household;
- by head education level;


- by household size (small household with 1-5 members and large household with more than 5 members); 

- by ownership of dwelling. 

Compare results. What can you conclude about the sensitivity of results to the choice of equivalence scales?  

4. Based on DIARY.sav file calculate the share of expenditures on each of following food categories: 

- Bread and cereals

- Meat

- Fish

- Milk, cheese and eggs

- Oils and fats

- Vegetables and fruits

- Sugar, jam, confectionery products

- Non-alcoholic beverages

- Alcoholic beverages
in the Total Food expenditures and compare the mean values for urban and rural population.

(C)   Creating consumption  aggregates 

The following SPSS syntax file is a simplifeied example of how to create the consumption by categories from the data fiule calle DIARY.SAV

*** This is a simlpe SPSS SYNTAX  example  on how to create the consumption aggtregates.

*** This file has been created for training purposes only .

GET

  FILE='DIARY.SAV' / KEEP HHID REGION TYPE PRODUCT  COICOP_G PAID_IN  QUANTITY.

***We want to change the 6 digit coicop codes from the the format XX.X.X.X.X to format  XXXXXX                       ***.

String coicop6 (A6).

comp coicop6= concat 

 (

     substr (coicop_g,1,2) ,

     substr (coicop_g,4,1) ,    

     substr (coicop_g,6,1) , 

     substr (coicop_g,8,1) ,

     substr (coicop_g,10,1)     ).

**The diary contains some income information coded under coicopp group 00, we want to drop them.

Select If  (substr( coicop6 ,1,2)<>'00') .

execute.

**** We want to assing the category to each expenditure and categporize thenin 4 categories.

** 1 (Non durable)   2  (Service)  3  (Semi Durable)  4  (Durable).

INCLUDE 'create_exp_type.sps'.

***This is a simple call for another Synatx , the details of calculation can be seen in it.

*we neet to create separate group for explicit (estimated by the owner of the house / appartment) 

* and implicit (rear lents paid by renters)  housing rents .

if PRODUCT="4211"  code2 ='13'    /* implicit rent*/   .

if PRODUCT="4112"  code2 ='14'   /* explicit rent*/   .

*COMMENT --- These codes taken from PREM ECAPOV stata codes.

if EXP_type = 4   code2 ='15'    /* Expenditures on Durable goods*/   .

ADD VALUE LABELS  code2 13 'implicit rents' 14 'explicit rents ' 15 'expenditures on durable goods' .

STRING code (A3).

COMP Code = CONCAT ('C' ,code2 ).

AGGREGATE

  /OUTFILE= *

  /BREAK=HHID Code

  /value 'Value of cumsumption in 15 days period' = SUM(PAID_IN).

exec.

* Reshape the long Diary to wide one and get the expenditures categories in a separate variables.

* the In current format of the aggregated diary the Houshold has many oservations, 

* we want to have one observation (row in the data file ) per one Household.

SORT CASES BY HHID code .

CASESTOVARS

 /ID = HHID

 /INDEX = code

 /GROUPBY = VARIABLE .

variable  labels 

  C01
'  Food and non-alcoholic beverages'

 /C02
'  Alcoholic beverages, robacco and narcotics'

 /C03
'  Clothing and footwear'

 /C04
'  Housing, water, electicity, gas and other fuels'

 /C05
'  Furnishing,household equipmentand routine maintenance of the house'

 /C06
'  Health'

 /C07
'  Transport'

 /C08
'  Communication '

 /C09
'  Recreatio and culture'

 /C10
'  Education'

 /C11
'  Restaurants and hotels'

 /C12
'  Miscellaneous goods and services' .

Variable labels

  C13
' Housing implict rents'

 /C14
' Housing Explicit rents'

 /C15
' Expenditures on Durable goods' .

recode all (sysmis =0).

* Calculating monthy nominal consumption aggregate.

* Monthy  consumption  excludes , health expenditures , rents and expenditures on durable goods.

Compute totconsm =  sum(C01,  c02 , c03 ,c04,c05,      c07,c08 ,c09, c10, c11,c12 )  * 2 .

Variable label totconsm  'Monthy  consumption '.

exec.

(D)   Adjusting with regional price differences 

The households living in the different regions need to spend different amount for getting the same bundle of products. If the nominal values of the per capita consumption of two different households living in the different regions are the same then we can assume that the household living in the region where the prices are cheaper is better off.

To make the consumption measure across the regions comparable we need to deflate the nominal per capita consumption to the so called regional price deflator.

